
Notes
About Me: 
I am the co-founder of Partner400, a firm specializing in customized education and mentoring services for IBM i (AS/
400, System i, iSeries, etc.) developers. My career in IT spans 45+ years including a 12 year period with IBM's 
Toronto Laboratory. 

Together with my partner Susan Gantner,  I devote my time to educating developers on techniques and technologies 
to extend and modernize their applications and development environments.  Together Susan and I author regular 
technical articles for the IBM publication, IBM Systems Magazine, IBM i edition, and the companion electronic 
newsletter, IBM i EXTRA. You may view articles in current and past issues and/or subscribe to the free newsletter at:  
www.IBMSystemsMag.com. We also write frequently for IT Jungle's RPG Guru column (www.itjungle.com). 

We also write a (mostly) monthly blog on Things "i" - and indeed anything else that takes our fancy. You can find the 
blog here: ibmsystemsmag.blogs.com/idevelop/   
Feel free to contact me any time:  Jon.Paris @ partner400.com 

Agenda
What exactly is JSON? 
• And why would I want to use it ? 

A Brief Introduction to JSON  
• Basic Syntax 

YAJL an Open Source JSON Tool 
• For both reading and writing JSON 

Introducing Scott Klement's IBM i port of YAJL 
• Scott has provided an RPG layer to simplify its usage 

Generating JSON with RPG 
• Simple Example 

Consuming JSON with RPG 
• And another simple example 

Other thoughts and options ...

JSON vs XML - And the winner is ...
JSON and XML are very similar in many ways 
• They provide a framework for describing data 

✦ Primarily for network communications 
• They are both self-describing  

✦ Meaningful field names are associated with the related data 
• Very popular for describing Web Service requests and responses 

✦ And Ajax calls in web applications 
• Supported by all major programming languages 

So why would I use JSON rather than XML ? 
• Other than in cases such as web services that demand the use of 

JSON 

Notes
JSON (JavaScript Object Notation) is fast becoming an essential technology in modern IBM i shops. From web 
services to browser interfaces and data exchange—it has seen a remarkable growth in usage over the last few 
years. In recent times IBM have even added direct JSON support to DB2 - a topic beyond the scope of this 
presentation however. 

JSON began basically as a replacement for XML in Ajax calls. That’s somewhat ironic when you consider that the "x" 
in Ajax actually stands for XML—“Asynchronous Javascript and XML" ...  But then the X in XML stands for 
eXtensible so a little irony is perhaps appropriate. 

JSON is "Better" Because ...
It is far more compact 
• Only 60 - 70% of the size of comparable XML 

✦ Shorter transmission overhead 
Its syntax is simpler 
• No namespaces 
• Simpler approach to escaping special characters 
• Only UTF-8 is supported 

✦ XML has to worry about multiple encodings 
Faster to decode 
• For example arrays have a specific syntax 

✦ Whereas arrays in XML are just repeats of an element

<customer><id>C12345</id><name>Charles Brown Inc.</name></customer>  

{ "id": "C12345", "name": "Charles Brown Inc." } 

Notes
XML had proven just too big and lumpy, requiring too much horsepower to parse in Web 2.0 type applications.  

JSON on the other hand is not only far more compact than XML, but because it is based on Javascript's own object 
data initialization definitions, a fast parser was already effectively built into every browser. So there was very little 
additional support needed in the browser.  

JSON also only uses UTF-8 encoding and so the multiple data encodings supported by XML do not have to be dealt 
with.  

With this three-fold advantage (compactness, ease of parsing, everything in UTF-8) it was hardly surprising that 
JSON rapidly replaced XML in most Ajax calls. From there it has gone on to become the vehicle of choice for web 
service requests and responses, and is making significant strides in the area of data interchange.

JSON Syntax
JSON data basically consists of name/value pairs 
• Although names are in some cases optional 

Individual elements within a series are separated by commas 
• If a name is present: 

✦ It must always be in quotes 
✦ And have a colon separate it from its associated value 

Values can be a ... 
• Number (integer or floating point) 
• String - always enclosed in quotes 
• Array ( Which can contain a mixture of data types) 
• Object (i.e. a collection of other stuff including objects) 
• Boolean true/false 
• or null    

{ "Name" : "Andrew Jones" , "Age" : 25 , "Position" : "Manager" } 

[ "Jan", "Feb", "Mar", "Apr", "May", "Jun", "Jul", "Aug", "Sep", "Oct", 
"Nov", "Dec" ]  

Object

Name/
Value Pair

Array

Notes
Basic rules for formatting JSON. 
  
JSON data consists of a series of name/value pairs with colons acting as the separator between the name and the 
value. Individual elements within a series are separated by commas and names must always be in quotes. So JSON 
data has this basic form: 

 "name1" : value_1, "name2" : value_2, ... "namen" : value_n 
  
The actual value portion of the name/value pair can be more than just a simple number or string. In fact it can be any 
of the following: 

Number (integer or floating point)  
String (enclosed in quotes) 
Array 
Object 
Boolean true/false 
or null. 

Arrays are delineated by square brackets ( "[" and "]" ). Unlike RPG arrays, which are limited to containing multiple 
elements of the same type and size, JSON arrays can contain any type of value, including another array. Which in 
turn can contain an array, which can ... You get the picture.  

JSON Syntax - Arrays
Arrays are delineated by square brackets 
• And can contain anything! 
• In the example below the array contains  

✦ Strings e.g "Nicola" 
✦ Numbers e.g. 39 
✦ And a null entry (she'd get mad at me if I gave away her age!) 

• The array entries could also have been represented by objects 
✦ e.g. { "name" : "Nicola", "age" : null } etc.  

[ "Jan", "Feb", "Mar", "Apr", "May", "Jun", "Jul", "Aug", "Sep", "Oct", 
"Nov", "Dec" ] 

{ "name":"Jon Paris", 
    "age":69, 
    "children": ["Nicola", null, "Steve", 39, "Rachel", 33, "Dave", 29], 
    "spouse": "Susan"  
}  

First entry in array Last entry in array

Notes
Probably the thing that is most surprising for an RPGer about JSON arrays is that the elements of an array do not all 
have to contain the same type of data. The first element could be a string, the second a number, and the third an 
object. If you already have some familiarity with languages like PHP this will not be a surprise - but if your 
background is RPG and CL it can seem a little odd. 
The closest thing we have in RPG is a Data Structure array and we'll look at that comparison in a moment. But even 
that doesn't have the same flexibility as JSON.

JSON Syntax - Objects
The term "Object" is not used in an OO sense  
• JSON Objects are just containers  

✦ Think of them as an RPG Data Structure 
They can contain any kind of JSON data 
• Including other objects 

✦ Think Nested Data Structure 
You can have arrays of objects as in this example ... 
• Or objects of arrays! 

Very flexible but can take some getting used to

{ "employees" : [ 
  { "firstName":"John", "lastName":"Doe", "age", 45 }, 
  { "firstName":"Anna", "lastName":"Smith", "age", 32 }, 
  { "firstName":"Peter", "lastName":"Jones", "age", 27 } ] 
}

Each array entry is 
an object

Notes
In my example I have an object which contains the item employees.   
employees is represented as an array ( [  ] ) 
Each array element is an object ( {  } ) 
Each of those objects contains three fields, firstname, lastname, and age. 
The value for each of the fields follows the colon ( : ) 

So - the first name of the third employee is Peter and he is 27 years old 
   

{ "employees" : [ 
  { "firstName":"John", "lastName":"Doe", "age": 45 }, 
  { "firstName":"Anna", "lastName":"Smith", "age": 32 }, 
  { "firstName":"Peter", "lastName":"Jones", "age": 27 } ] 
} 

As an RPG programmers this can be a little difficult to get your head around to start with.

Comparing a JSON Object with an RPG DS
Let's compare a JSON Object with an RPG Data Structure 
• The DS employees in this example 
• This is the kind of structure you will use to process JSON using the 

new DATA-INTO opcode 
Remember - JSON arrays can contain multiple types of data 
• So they may not always have a simple RPG equivalent 

 
{ "employees" : [ 
  { "firstName":"John", "lastName":"Doe", "age", 45 }, 
  { "firstName":"Anna", "lastName":"Smith", "age", 32 }, 
  { "firstName":"Peter", "lastName":"Jones", "age", 27 } ] 
} 

dcl-ds employees  Dim(3)  Qualified; 
   firstName  Char(20); 
   lastname   Char(20); 
   age        Int(5); 
end-ds; 

Notes
As you will discover when you work with JSON, sometimes its "relaxed" rules (compared with the rigid rules of XML) 
can make it harder to work out what the corresponding RPG DS should look like. 
For example, a JSON array, like the simple one below that I referenced earlier, may not have a name. 

[ "Jan", "Feb", "Mar", "Apr", "May", "Jun", "Jul", "Aug", "Sep", "Oct", 
"Nov", "Dec" ] 

In XML it would have likely been coded something like this: 

<MonthNames> 
  <Name>Jan</Name><Name>Feb</Name>...<Name>Dec</Name> 
</MonthNames> 

Which maps easily to a DS like this: 

dcl-ds MonthNames; 
  Name  char(9)  Dim(12); 
end-ds; 

The connection is far more obvious than in JSON where you may have to make up the names yourself. 

Enough of all that Syntax Stuff ...
The YAJL Library 
• YAJL = Yet Another JSON Library 

✦ Aren't meaningful names wonderful!  
• Originally created by Mozilla's Lloyd Hilaiel 

✦ He now works at Facebook  
• Open Source 
• Written in C and very, very fast 
• Ported to IBM i (ILE C) by Scott Klement & David Russo 

Scott also added direct RPG interfaces (YAJLR4) 
• These simplify calling YAJL 
• And handle the UTF-8/EBCDIC translation for you 

You can download the whole package at: 
• www.scottklement.com/yajl/

Notes
Scott Klement's port of YAJL is a great tool both for creation and consumption of JSON.  
You can download the latest version of YAJL free of charge from Scott's site (www.scottklement.com/yajl/) and the 
install process is very simple and well documented. Did I mention that YAJL is very fast? Blazingly fast!  

A of course Scott has done his usual stellar job of "RPGizing" the API calls to make it easy to use. For example, his 
RPG interfaces automatically take care of translating between EBCDIC and UTF-8. As noted on the chart there are 
times when this is actually unhelpful as you'll see later, but since you can always invoke the underlying JAJL APIs 
this is not a problem. 

As noted,YAJL stands for "Yet Another Javascript Library"—not the most useful of acronyms! This package offers an 
API driven approach for JSON generation that is highly maintainable and easy to understand.

Creating JSON with YAJL
Our example is designed to create customer information 
• Here's what a real version of the generated file might look like 

✦ This is the "pretty" form - the notes page shows a production version 
• Our demo program though will just generate fake data 

{  "Customers": { 
      "Customer": [ 
         { 
            "Id": "G011739", 
            "Name": "Langosh, Borer and Homenick", 
            "Address": { 
               "City": "Danykafurt", 
               "State": "NV" 
            } 
         }, 
         { 
            "Id": "E791196", 
            "Name": "Denesik, Kessler and Rolfson", 
        ... 

Notes
Like XML, JSON is not dependent on whitespace and so the document would likely look far more like this in practice: 

{ "Customers": { "Customer": [ { "Id": "G011739", "Name": "Langosh, Borer and 
Homenick", "Address": { "City": "Danykafurt", "State": "NV" } },  { "Id": "E791196", 
"Name": "Denesik, Kessler and Rolfson", 

... 

Creating JSON with YAJL - The APIs
The APIs that this example uses are: 
• yajl_genOpen() 

✦ This begins the generation process  
• yajl_beginObj() and yajl_endObj() 

✦ These begin and end a JSON Object 
• yajl_addChar 

✦ Adds a character value pair (i.e. the name and its value) 
• yajl_beginArray() and yajl_endArray() 

✦ Used to start and end an array 
• yajl_saveBuf() 

✦ Save the contents of the buffer to an IFS file 
- You would replace this with yajl_copyBuf() if you were going to take the 

output and send it directly to (say) a web service 

This code is based on an article I wrote for IBM Systems Mag 
• ibmsystemsmag.com/ibmi/developer/rpg/first-steps-json/

Notes
The code sample we are about to look at is taken from an article that I wrote for IBM Systems Mag. You'll find the 
article here: 
ibmsystemsmag.com/ibmi/developer/rpg/first-steps-json/ 

And you can copy/paste the full source code for this program from this link: 
partner400.com/examples/jsonyajl.htm 

Where you will also find a second example that generates the employee data discussed in the article and a small of 
which was shown earlier in the presentation. 

P.S. Both links above should be "clickable" in the PDF. 

The RPG Code - Part 1

 ctl-opt  DftActgrp(*No)    BndDir('YAJL') 
          Option(*SrcStmt)  DecEdit('0.'); 

/copy yajl/qrpglesrc,yajl_h 

 dcl-s  i       Int(5); 
 dcl-s  wait    Char(1); 
 dcl-s  errMsg  VarChar(500) Inz; 

 // Main Process Begins ... 
       
 yajl_genOpen(*On);  // Turn on "pretty" mode 

 yajl_beginObj();  // Start the root object 

   yajl_beginObj('Customers'); 

     yajl_beginArray('Customer'); 

This is a great 
development option. It 

makes it easier to read the 
JSON being generated

Notes
Processing begins with a call to the yajl_genOpen() procedure to begins the JSON generation process.  
The parameter *On tells YAJL to produce "pretty" output. We would normally do this while in the development phase 
to make it easier to diagnose problems in the generation. Once the program is working and ready to roll into 
production we would simple change this parameter to *Off and JYAJL would then produce a compact non-formatted 
stream.  

Next yajl_beginObj() - passing no parameter signals the API to start a new JSON object that will contain the entire 
document - in XML terms this would be the root object.  

This is followed by a second call to the same API to begin the object named "Customers".  

The customers object consists of an array of customer details so the next thing to do is to start the customer array by 
calling yajl_beginArray(). 

The RPG Code - Part 2

     yajl_beginArray('Customer'); 

     // This code is just "faking" the data so a simple loop 
     //   is used as a substitute for normal data retrieval 

       For i = 1 to 3; 

         yajl_beginObj(); 

           yajl_addChar('Id': %Char(i)); 
           yajl_addChar('Name': %Char(i)); 

           yajl_beginObj('Address'); 
             yajl_addChar('City': %Char(i)); 
             yajl_addChar('State': %Char(i)); 
           yajl_endObj(); 

         yajl_endObj(); 

       EndFor;

Notes
This is just an example program that I want you to be able to copy and "play" with it on your own system without 
having to worry about creating data files, etc. So I am simply using a For loop to generate fake data for three 
customers. The actual data values are simply a character (%Char) version of the current loop index. Obviously in a 
real working version the data would likely be supplied by an SQL query or RPG I/O operations.  

Each element in the customer array is an object, so we start each loop wth a call to yajl_beginObj() and follow that 
by using yajl_addChar() to add the character variables for the Id and Name fields. 
The address however is an object that contains city and state so once again we call yajl_beginObj() followed by the 
calls to yajl_addChar() to add the required fields. 

We then make two consecutive calls to yajl_endObj(). The first terminates the address and the second the array 
element. 

We then loop back to process the next customer. 

The RPG Code - Part 3

       EndFor; 

       yajl_endArray(); 

     yajl_endObj();   // End Customer 

   yajl_endObj();   // End root 

   // Change this to reflect where you want to store the file 
   yajl_saveBuf('/Home/JSONEX2.json': errMsg ); 

   if errMsg <> ''; 
     Dsply 'Ooppppssss - problem!' ' ' Wait; 
   EndIf;

Notes
Once all the data has been output (i.e. the For loop ends) yajl_endArray() is called to end the customer array. Then 
two consecutive calls are made to yajl_endObj() to wrap up the customer and root objects. 
The JSON document is now complete and ready to do whatever we wish with it. In the case of this simple example, 
and to make it easy to see the results of running the program, I have simply saved the whole buffer to an IFS file. 
Change this name to anything you want. 

Notice that in addition to providing the file name, I also supplied an error message field (errMsg). In the event of a 
failure while trying to write to the IFS, this would contain an appropriate error message; otherwise it will be an empty 
string. This possibility is tested and an error message displayed if there is a problem. 

Checking Your Work
Make sure you validate what you build 
• It helps avoids arguments with the people processing the JSON ! 

Output the JSON to a file 
• Even if that is not your ultimate objective 

Use the option to generate "pretty" JSON 
• You request it via the yajl_genOpen() 
• It makes it much easier to spot silly mistakes in nesting levels 

Run the output through a JSON validator 
• For example: 

✦ JSONLint ( jsonlint.com )  
✦ FreeFormatter.com. ( freeformatter.com/json-validator.html )

Notes
When developing JSON documents it is a good idea to write the results to the IFS, as shown in this example. Even if 
you will not be doing this in production it gives you an easy way to view the generated results. More importantly it 
allows you to easily check the results with a JSON validator. That way you can be sure your document is valid before 
you start sending it out and getting into arguments with other developers as to whether the generation of the JSON 
or its subsequent processing is the cause of problems.  

There are many validators out there on the web including: 
JSONLint ( jsonlint.com )  
and FreeFormatter.com. ( freeformatter.com/json-validator.html ) This one has the added advantage of allowing you 
to upload a file for checking.

Notes on the YAJL APIs
Other common YAJL APIs are: 
• yajl_addBool() which adds a boolean (true/false) value 
• yajl_addNum() which adds a numeric value. 
• yajl_addNull() which, not surprisingly, adds a null 
• There are others including one that adds a stream file's content 

Many of these APIs are a little different than the usual 
• For example I used yajl_addChar(); like this: 

✦ yajl_addChar( 'Id' :  %Char(i) ); 
• Two parameters - the first is the field name, the second the value 
• But I could have just passed an unnamed value to be added  

✦ e.g. yajl_addChar( %Char(i) ); 
• As you can see the actual value now becomes the first parm 

It is all perfectly logical 
• Just different from what we are used to

Notes
There are other APIs available, including some designed to accommodate huge character strings that exceed RPG's 
limits. They are not the ones that you will use frequently and so I'm leaving you to your own devices to study them if 
you need them. The YAJL prototypes do a pretty good job of explaining them. 

"Different" ? 

As noted the difference is all perfectly logical, but it is perhaps not how we would normally expect RPG 
subprocedures to work.  
If we have an optional parameter it is normally the last one in the sequence. So to allow for the field name to be 
omitted we would normally have coded the value parameter as the first and the name as the second with a *NoPass 
option. Of course that would not logically map to the resulting JSON which would always have the variable name 
first. 

Consuming JSON with YAJL
YAJL provides two ways to read JSON data 
• Event-based - similar to XML-SAX in RPG and  
• Tree-based (like DOM in XML) APIs  

The tree method is generally easier to use 
• So that will be my focus 

YAJL loads the JSON into memory and organizes it hierarchically 
• Your code then "walks" through the branches of the tree 

I am using yajl_stmf_load_tree() to load the JSON from a file 
• There are similar alternatives to load from variables etc. 

The variable root is known as a node 
• It will be used as the base for drilling down into the tree 

This example is based on an article I wrote for IBM Systems Mag 
• ibmsystemsmag.com/ibmi/developer/rpg/json-data-with-yajl/

root = yajl_stmf_load_tree ( '/Home/Customers.json' 
                                : errMsg ); 

Notes
The method my example uses is the tree parser. This works on the document as a whole and allows you to basically 
reference data elements by name. You’ll see that when we look at the code. 

The basic methodology that is used is to first load the "tree" with the entire JSON document. 

Subsequently you obtain a node locator—an identifier or "key" if you want to think of it that way—for each level of the 
document you want to process. That locator is then used to drill down to the next level.  

There really is no RPG equivalent, but the effect is similar to if you did a serial READ of a file and then used data 
from that to CHAIN out to another file from which you used a field to CHAIN to another file from which ... 

If you are familiar with the concept of linked lists you will see those kinds of concepts at play here. 

As you'll see, we'll begin by asking YAJL to supply us with the locator for the Customers element, and then use that 
to locate the nodes for the individual customer array elements. These will then be used to access the individual fields 
that make up customer data. It sounds a lot more complicated than it actually is, as you'll see when we look at the 
code. 

Sample JSON to be Processed
I am processing a simplified version of the customers document 
• This is what it looks like ...

  { "Customers": [ 
       {  
         "ID": 12345, 
         "Name": "Paris", 
         "Street": "Main Street", 
         "City": "Jasontown", 
         "State": "CA", 
         "Zip": "12345" 
       }, 
       {  
         "ID": 23456, 
         "Name": "Rich", 
         < ... data for multiple customer entries omitted ... > 
  
         "Zip": "98765" 
       } 
     ] 
  }

Notes
The document starts with an array named Customers.  

This is followed by the data for the first customer which begins with an opening "{". The individual elements (ID, 
Name, etc.) follow and are separated by commas. The final item for the first customer is terminated by a closing "}". 

This document highlights a difference between XML and JSON.  In XML we would be forced to name the repeating 
element (i.e. the customer data) in order to be able to group its related fields. JSON does not require names for 
everything, so simply grouping items into an object as we have done here is sufficient. 

And this is its Document Tree

Once you have the node for a given level in 
the tree you can access the next level down

Document 
Root

Array Element 
Node

State

Customers 
Node

ID Name Street City Zip

Notes
I had trouble placing this chart. This tree does not come into existence until after YAJL has loaded the document into 
memory - which we will see in the next couple of charts.  

Hopefully having it here will help you to understand the process of walking down (up?) the tree to obtain the content 
of the fields at the different levels.

Consuming JSON - Part 1

 /copy yajl/qrpglesrc,yajl_h 

  // Node locators 
  dcl-s  root           Like(yajl_val); 
  dcl-s  addressNode    Like(yajl_val); 
  dcl-s  customersNode  Like(yajl_val); 
  dcl-s  customerNode   Like(yajl_val); 
  ... 

  dcl-ds  customer  Dim(99)  Inz  Qualified; 

    id       Zoned(5); 
    name     Char(40); 
    street   Char(40); 
    city     Char(30); 
    state    Char(2); 
    zip      Char(5); 

  end-ds; 

Start by copying in the 
YAJL prototypes etc.

Actually these are 
pointers but these don't 
look as scary!

Data extracted from the JSON 
will be stored here for 
subsequent processing

Notes
The program starts by copying in the prototypes for the YAJL routines.  
Next come the definitions for our node locators. In order to make the code a little simpler to understand I have 
defined a separate node locator for each and every one of the elements I will be processing. Each is defined as 
being like a yajl_val. These are actually pointers, but Scott wisely "masked" this to protect the pointer phobic. 

I could have used the same node locator multiple times but this way it makes the code a little simpler to follow I think.

Consuming JSON - Part 2

  root = yajl_stmf_load_tree ( '/Home/Customers.json' 
                               : errMsg ); 

  // If an error was found then report and exit 
  if errMsg <> ''; 
    Dsply 'Ooppppssss - file load problem!'; 
    // Add appropriate error handling ... 
    Return; 
  EndIf; 

  customersNode = YAJL_OBJECT_FIND( root: 'Customers' ); 

  elements = YAJL_ARRAY_SIZE( customersNode ); 

  If elements > %Elem(customer);  // Too many to handle? 
    Dsply ('Can only process ' + %Char( %Elem(customer) ) + 
           ' customers - File contains ' + %Char(elements) ); 
    Return; 
  endif; 

Load the tree 
from the file

Error could be caused by 
file not found, or JSON 
syntax error, or ...

Simple error check to ensure 
we don't blow the array limit

Notes
Since this particular JSON document is stored in the IFS I have used the API yajl_stmf_load_tree() API to load the 
document identified by the first parameter, into YAJL's document tree. If the load is successful, the API returns the 
node locator for the root element. If an error were encountered then a message would be placed in the second 
parameter ( errMsg ). Possible errors include the file being missing, or otherwise inaccessible, or syntax errors 
encountered when parsing the JSON data. 
Next I test the error message to see if any problems were encountered and take appropriate action should that be 
the case.  
Now the real work begins with using YAJL_OBJECT_FIND to locate the Customers object. The first parameter is the 
node locator for the part of the tree that we want to search. In this it is the root for the entire document as supplied by 
the load tree routine.  
You might be wondering what would happen if we failed to locate the Customers object. Good question - but I'll defer 
the answer to later when I will discuss the handling of optional elements. 
The next part demonstrates a really useful YAJL function: YAJL_ARRAY_SIZE. This returns the number of elements 
found in the array identified by the node locator passed to it. In this case we are passing the locator to the 
Customers array.  
The subsequent code is used just to demonstrate a way to defend against any problems that could be caused by 
RPG's fixed array size. In this program I allowed for 99 customers in the array. This code makes sure that we defend 
against any attempt to process more than that number. 

Consuming JSON - Part 3

  c = 0; 
       
  Dow YAJL_ARRAY_LOOP( customersNode: c: customerNode ); 

     idNode = YAJL_OBJECT_FIND( customerNode: 'ID' ); 
     customer(c).id = YAJL_GET_NUMBER( idNode ); 

     nameNode = YAJL_OBJECT_FIND( customerNode: 'Name' ); 
     customer(c).name = YAJL_GET_STRING( nameNode ); 

     streetNode = YAJL_OBJECT_FIND( customerNode: 'Street' ); 
     customer(c).street = YAJL_GET_STRING( streetNode ); 

                 // Repeat for all fields ... 

  EndDo; 

Clear the loop index
customerNode is set 
for each new array 

node found

Obtain the 
node then 
extract the 

value

Use the loop 
index as the 

array subscript

Notes
I have a locator which points to the Customer array (customersNode) and can use it to process each of the individual 
customer elements in turn. This is done by using YAJL_ARRAY_LOOP. This returns an indicator which will be *On if 
an element is found, and *Off when there are no more elements left to process. That allows us to use a DOW loop to 
easily iterate through all of the elements in the array.  
The first parameter (customersNode) is the node we wish to traverse, the second (c) is a counter that identifies the 
array element number to start from. More on this in a moment. The third parameter (customerNode) will be used to 
return the node locator for each array element (i.e., customer) that is found. 
Two points about the counter (c): The first is that it’s automatically incremented each time the function is called. This 
allows us to use it as an array index when loading the extracted data into the array DS as you will see in a moment. 
The second is that the function uses the counter to determine which element to retrieve. That is the reason why I had 
to clear the counter before entering the loop. 
Using the retrieved customerNode locator we can now extract the component fields. We begin by using 
YAJL_OBJECT_FIND to locate the ID field's node, and then extract its value using YAJL_GET_NUMBER. The result 
is loaded directly into the corresponding element of the customer DS array using the counter as the index. This 
process is then repeated for each of the customer fields. Note that for character fields we used YAJL_GET_STRING 
to perform the corresponding data extraction. 
That's all there is to it really. 

Alternative Approaches - 1
Requesting the value for each field in turn is fine 
• But only if all fields are present 

What if one were missing ? 
• e.g., an optional field 

Two basic approaches 
• Test the node returned by YAJL_OBJECT_FIND() to see if it is null 

✦ Since the node is actually a pointer this works just fine 
- See the example below 

• Use YAJL_OBJECT_LOOP() to loop through all fields 
✦ And assign values based on field name 

- Example on the next page

 idNode = YAJL_OBJECT_FIND( customerNode: 'ID' ); 
         If idNode <> *null; 
            customer(c).id = YAJL_GET_NUMBER( idNode ); 
         Else; 
            // Take appropriate error action 

Notes
The first method is to test the node returned by YAJL_OBJECT_FIND() to see if it is null. Remember that underneath 
it all the node values are pointers, and the pointer will be null if no object was found. The word "*null" works well here 
to making the test pretty much self-documenting. 
If ID were a compulsory field, which given its name is likely, then this is probably an appropriate way to handle it. But 
what if there are optional fields involved in the document? There has to be an easier way than coding a test for each 
and every field.  

And there is by using YAJL_OBJECT_LOOP() as we'll see on the next chart. 

Alternative Approaches - 2
Using YAJL_OBJECT_LOOP() 
• Very similar in operation to YAJL_ARRAY_LOOP() 

It returns the name (key) and node for each object it finds 
• And also auto-increments the item counter (i) 

It replaces YAJL_OBJECT_FIND() used in the earlier example

     i = 0; 

     Dow YAJL_OBJECT_LOOP( customerNode: i: key: node ); 
        Select; 
        When key = 'ID'; 
           customer(i).id = YAJL_GET_NUMBER( node ); 
        When key = 'Name'; 
           customer(i).name = YAJL_GET_STRING( node ); 
        // .... and all the other fields 
        When key = 'Zip'; 
           customer(i).zip = YAJL_GET_STRING( node ); 
        EndSl; 
     EndDo; 
 

Don't forget to test for 
compulsory elements if 

using this method.

Notes

When using YAJL_OBJECT_LOOP() you begin by setting the item counter (i) to zero. It will subsequently be 
incremented as each object is retrieved. This function requires four parameters. The first is the node locator from 
which we are to extract the objects - customerNode in our case. The second is the item counter. The third is a 
character field into which the function will place the key value (i.e. the object's name) and it will place the actual 
locator node in the fourth parameter. Like YAJL_ARRAY_LOOP this function returns an indicator that is used to 
control the DOW loop. 
You can then code a SELECT and test for the individual field names in the WHEN clauses. In practice, when using 
this approach, we would almost certainly need to add logic to ensure that compulsory items were in fact present. We 
would also need to ensure that appropriate default values were set for the optional fields. For example by ensuring 
that the optional character fields in the array DS were blank. 

One Last (Important) Feature
yajl_stmf_load_tree()  
• Used at the start to load the tree into memory 
• Returns the node of the root element 

✦ It is vital to keep a copy of this node so that you can release the 
memory later but using ... 

yajl_tree_free() 
• Releases the allocated memory 
• It will also free any and all child nodes based on it 

✦ So make sure you do not refer to any of those nodes after using it 

Notes
This is just a question of "cleaning up" and releasing the dynamic memory back to the system. Failure to do so in (for 
example) a batch job would probably not cause any issues as the memory would be cleaned up at the end of the job. 
In an interactive job or, worse still, a never ending web server job, failure to tidy up could eventually cause issues. 

Better to be a good corporate citizen and tidy up after yourself as you go. 

Other Thoughts And Options
Other options for generating JSON include: 
• powerExt 
• Any template-based system such as CGIDEV2 uses 
• SQL's new JSON generation capabilities 
• And of course just stringing stuff together 

Options for consuming JSON 
• powerExt 
• RPG's new DATA-INTO opcode with the supplied JSON parser 

✦ Or use DATA-INTO with the new YAJL-based parser just added to the 
package by Scott 
- It is far more flexible then the IBM supplied sample JSON parser 

• SQL's new JSON consumption capabilities

Notes
I've mentioned here only the free options and those that can be used directly from within an RPG program. There are 
also a number of commercial options available. 

Plus of course PHP, Java, and Python, and node.js all provide JSON generation and consumption capabilities. 

The End ... 

Any Questions ??

Notes
In addition to my magazine articles referenced earlier there are a couple of other spots on the web where you can 
find more information on YAJL.  

In addition to the examples that Scott Klement includes on his download pages, Brad Stone has also written a 
number of blog posts describing his explorations with YAJL.  Here are the links: 

Scott's YAJL homepage: scottklement.com/yajl/  
You'll find links to the main download plus a number of examples 

Brad Stone's fieldexit.com 
Building JSON: fieldexit.com/forum/display?threadid=195 
Parsing JSON part 1: fieldexit.com/forum/display?threadid=197 
Parsing JSON part 2: fieldexit.com/forum/display?threadid=199 
Parsing JSON part 3: fieldexit.com/forum/display?threadid=201 
Parsing JSON part 4: fieldexit.com/forum/display?threadid=204

